
!- Note: Curly brackets {} represent special screen code characters (BASTEXT PETSCII Tokens) 

!- For BASTEXT reference see: https://www.c64-wiki.com/wiki/PETSCII_Codes_in_Listings 

!- For checksum tool see    : https://www.c64-wiki.com/wiki/F64Summer 

 

CSUM  

1242 10 rem ******************************* 

03FC 20 rem ***         yahtzee         *** 

4B59 30 rem ***   (c) 2020 by @romwer   *** 

9474 40 rem *** roman.werner@gmail.com  *** 

4319 50 rem ******************************* 

39AF 60 rem 

C797 70 rem **************************** 

E89E 80 rem *** initialize variables *** 

8042 90 rem **************************** 

B5B8 100 dimc(15,3):dimp$(17) 

5BE7 110 d$="{home}{down*23}":r$="{down}{left*5}":b$="{space*5}"+r$:t$="playerround " 

88DE 120 l$="&{reverse on}{space*3}{reverse off}'"+r$:m$="{white}{light green}" 

D95D 130 dc$="{white}#$$$%"+r$+l$+l$+l$+"()))*{left*4}{up*3}{black}{reverse on}" 

4CDB 140 dc$(1)="{down}{right}!":dc$(2)="{space*2}!{down*2}{left*3}!":dc$(3)="{down*2}!{up}!{up}!" 

1D74 150 dc$(4)="! !{down*2}{left*3}! !":dc$(5)="! !{down}{left*2}!{down}{left*2}! !" 

D0BB 160 dc$(6)="! !{down}{left*3}! !{down}{left*3}! !" 

0879 170 rem ************************************************** 

F6F8 180 rem *** initialize environment (chars and sprites) *** 

EF93 190 rem ************************************************** 

C22B 200 print"{clear}"tab(14)"{black}please wait":v=53248:s=54272 

4989 210 pokev+21,0:pokev+32,5:pokev+33,5:pokev+34,2:pokev+35,1:pokev+36,0 

FA0C 220 fori=stos+24:pokei,0:next:pokes+8,6:pokes+15,4:pokes+24,15 

1A14 230 gosub1730:pokev+24,peek(v+24)and240or14:pokev+17,peek(v+17)or64::gosub2030 

5511 240 rem ************************* 

57DF 250 rem *** show title screen *** 

13DF 260 rem ************************* 

EA9F 270 fori=0to15:forj=0to3:c(i,j)=0:next:next 

88B4 280 a=fre(.):p=0:r=0:f=5:t=0:p$(0)="":fori=1to5:l(i)=0:next 

5142 290 print"{clear}"tab(16)"{black}yahtzee":pokev+21,15 

7E12 300 gosub1400 

EE04 310 printleft$(d$,11) 

9A67 320 printtab(10)"{white}welcome to yahtzee":print 

0974 330 printtab(14)"{yellow}f1{light green} 1 player" 

8343 340 printtab(14)"{yellow}f3{light green} 2 player" 

2119 350 printtab(14)"{yellow}f5{light green} 3 player" 

D99D 360 printtab(14)"{yellow}f7{light green} 4 player" 

2109 370 printtab(15)"{down}{yellow}p{light green}ip color" 

5F31 380 gosub1620 

C602 390 ifk=80thengosub1500 

4ADD 400 ifk<133ork>136goto380 

3A54 410 z=k-133 

CE96 420 j=24:gosub1680 

9DCA 430 pokev+21,0:fori=1to9:print"{home}{right}{delete}";:wait53265,128:next 

BE2B 440 rem ***************** 

2548 450 rem *** main loop *** 

DC2F 460 rem ***************** 

5276 470 f=5:g=0:t=1:fori=1to5:l(i)=0:next:onz+1goto600:t=0:ifp>zthenp=0 

3522 480 gosub1040 

7340 490 gosub1620 

C23A 500 ift>0andt<3thenifk>48andk<54thengosub680:goto490 

06B6 510 ift>0thenifk>64andk<78thengosub750:onggoto570,270:goto490 

1D08 520 ifk=80thengosub1500 

4370 530 ifk<>32ort=3goto490 

ED05 540 ift=0andz>0thenpoke781,21:sys59903:poke781,23:sys59903 

33AA 550 rem *** prepare rolling the dice... *** 

5338 560 f=5:fori=1to5:f=f-l(i):next:iff=0goto490 

2065 570 t=t+1:ift=4thenp=p-(z>0):j=24:gosub1680:goto470 

096C 580 fori=1to5:ifl(i)=0thenprintleft$(d$,20)tab(2+i*5)b$b$b$b$b$"{space*5}"; 

BD6C 590 next:printleft$(d$,22)" "b$b$b$;left$(d$,22)tab(33)b$b$b$; 

7C05 600 j=19:gosub1680:pokev+21,15: 

7301 610 fori=0tof:pokes+1,rnd(1)*10+1:fork=1to2:forj=0to2:pokes+4+7*j,129:next 

628E 620 forj=0to2:pokes+4+7*j,0:next:next:forj=1to5:next:next 

0E7B 630 gosub1400 

8877 640 goto480 

5AA7 650 rem ***************** 

1893 660 rem *** lock dice *** 

69D4 670 rem ***************** 

B232 680 d=k-48:l(d)=1-l(d):printleft$(d$,20)tab(4+d*5); 

AA3D 690 ifl(d)=1thenprint"{light green}>"; 

9477 700 ifl(d)=0thenprint"{yellow}"chr$(k); 

51BD 710 return 

CFCF 720 rem ********************** 

3E98 730 rem *** score category *** 

A4DD 740 rem ********************** 

51EB 750 d=k-65:d=d-2*(d>5):ifpeek(1125+d*40+p*5)<>32thenreturn 

1DA8 760 g=1:n=0:x=0:y=0:b=19+p*5:t=3 

B84B 770 ifd<6thene=d+1:fori=1to5:n=n-e*(x(i)=e):next:goto860 

0B42 780 fori=1to6:d(i)=0:next:fori=1to5:d(x(i))=d(x(i))+1:n=n+x(i):next 

D14C 790 ond-7goto800,810,820,830,840,850,860 

0CDC 800 fori=1to6:x=xor(d(i)>2):next:n=n*-x:goto860 

B115 810 fori=1to6:x=xor(d(i)>3):next:n=n*-x:goto860 

1490 820 fori=1to6:x=xor(d(i)=2):y=yor(d(i)=3):next:n=25*-(x<0andy<0):goto860 

9DA4 830 fori=1to6:x=(x-(d(i)>0))*-not(x<4andd(i)=0):next:n=30*-(x>3):goto860 

BA44 840 fori=1to6:x=(x-(d(i)>0))*-not(x<5andd(i)=0):next:n=40*-(x>4):goto860 

3280 850 fori=1to6:x=xor(d(i)=5):next:n=50*-x 

3186 860 l=d:gosub1560:printleft$(d$,3+d)tab(b)"{light green}!{white}"right$(n$,2) 

7805 870 ifd<6thenfori=0to5:n=n+c(i,p):next:l=6:gosub1560:printleft$(d$,9)tab(b)m$n$ 

5694 880 ifc(6,p)>62thenifc(7,p)=0thenn=35:l=7:gosub1560:printtab(b)"{white}"n$ 

29F4 890 fori=6to14:n=n+c(i,p):next:l=15:gosub1560:printleft$(d$,18)tab(b)m$n$ 

778D 900 poke781,19:sys59903 

3B9A 910 ifp=zthenr=r+1:ifr=13theng=2:goto970 

735D 920 print"{down*3} "b$b$b$"{up*3}"tab(33)"space"r$"=next"r$mid$(t$,-(z=p)*6+1,6) 

B278 930 gosub1620:ifk=80thengosub1500 

65A3 940 ifk<>32goto930 

303C 950 return 

FF20 960 rem *** show game over *** 

4C94 970 poke1123+d*40+p*5,32:print"{down*3} game "r$b$"over {up*2}"tab(33)"press"r$b$"space" 

FEBE 980 onz+1goto930:n=c(15,0):fori=1toz:ifc(15,i)>nthenn=c(15,i) 

93B1 990 next:print"{home}{yellow}";:fori=0toz:printtab(18+i*5)left$("@",-(c(15,i)=n));:next 

EC08 1000 goto930 

3B9E 1010 rem ************************ 

ECAF 1020 rem *** show score table *** 

108E 1030 rem ************************ 

0C0E 1040 pokev+21,0:print"{home}{down}" 

89E1 1050 printtab(8)"{white}ones [ {yellow}a" 

B8BC 1060 printtab(8)"{white}twos £ {yellow}b" 

49EC 1070 printtab(6)"{white}threes ] {yellow}c" 

E472 1080 printtab(7)"{white}fours ^ {yellow}d" 

E4D2 1090 printtab(7)"{white}fives {arrow left} {yellow}e" 

718B 1100 printtab(7)"{white}sixes : {yellow}f" 

875F 1110 printtab(6)"{light green}subtotal" 

5EFB 1120 printtab(5)"{light green}63+ bonus" 

48A6 1130 print"{space*3}{white}3 of a kind {yellow}g" 

A509 1140 print"{space*3}{white}4 of a kind {yellow}h" 

B974 1150 print"{space*4}{white}full house {yellow}i" 

3AD1 1160 print"{white}small straight {yellow}j" 

77FA 1170 print" {white}long straight {yellow}k" 

AB5B 1180 printtab(7)"{white}yahtzee {yellow}l" 

50FC 1190 printtab(8)"{white}chance {yellow}m" 

8B99 1200 printtab(9)"{light green}total" 

3088 1210 print"{home}"tab(16)"{white} "; 

EF10 1220 ifz>0thenfori=0toz:printmid$(m$,2+(p=i),1)+"{space*2}pl"+mid$("1234",i+1,1);:next 

F05F 1230 ifp$(0)<>""thengoto1300 

D9D9 1240 p$="{light green}#;":onz+1goto1250:fori=1toz:p$=p$+";;;;<":next 

7A99 1250 p$(0)=p$+";;;;%":fori=0to15:p$="{light green}& " 

19DC 1260 forj=0toz:ifi=6ori=15thenp$=p$+"{light green}{space*2}0":goto1280 

9CF0 1270 p$=p$+"{white}{space*3}" 

EC51 1280 p$=p$+" {light green}'":next:p$(1+i)=p$:next 

A31F 1290 p$(17)="{light green}("+left$("{)*20}",(z+1)*5)+"*" 

9B85 1300 print:fori=0to17:printtab(17)p$(i):next 

2DA9 1310 ift>0andt<3thenford=1to5:printtab(4+d*5);:k=d+48:gosub690:next 

B56A 1320 ift>0thenprintleft$(d$,22)" {light green}a-m ="r$"score"r$"categ{up*5}" 

88A4 1330 ift>0andt<3thenprintleft$(d$,22)tab(33)"{light green}space"r$"=roll"r$"again{home}" 

AF59 1340 ift=0thenprintleft$(d$,22)tab(16)"{white}player"str$(p+1) 

F881 1350 ift=0thenprinttab(6)"{down}{light green}press space to roll the dice{home}" 

F44A 1360 return 

AE69 1370 rem *********************** 

24CD 1380 rem * dice roll animation * 

BA52 1390 rem *********************** 

F399 1400 fori=1to5:w(i)=(int(rnd(.)*4)+8)*-(l(i)=0):next 

D360 1410 fori=1to5:w=w(i):onw+1goto1450:w=w-1+(w>4):w(i)=w:ifw=0thenf=f-1 

9D6A 1420 ifw=0thenift=0thenx=i:goto1440 

9D34 1430 x=int(rnd(.)*6)+1:ifx=x(i)orx+x(i)=7goto1430 

24A8 1440 x(i)=x:printleft$(d$,19-w)tab(2+i*5)b$b$dc$mid$(c$,x,1)dc$(x) 

7560 1450 next:forj=1to400-f*80:next:iffgoto1410 

32EA 1460 return 

CA7A 1470 rem ************************ 

B749 1480 rem *** toggle pip color *** 

CCE4 1490 rem ************************ 

3496 1500 c=1-c:c$=left$("{red}{blue}{green}{purple}{pink}{cyan}",c*6):fori=1to5:w(i)=1:next 

2A42 1510 fori=1to5:printleft$(d$,21)tab(2+i*5)dc$mid$(c$,x(i),1)dc$(x(i)):next 

6D4A 1520 return 

F2DE 1530 rem ********************************************************** 



 

 

CD38 1540 rem *** update score table element (in value+string array) *** 

60BB 1550 rem ********************************************************** 

2337 1560 c(l,p)=n:n$=str$(n):n$=right$("{space*2}"+mid$(n$,2,len(n$)-1),3):n=0 

1D19 1570 p$=p$(l+1):p$(l+1)=left$(p$,4+p*7)+n$+mid$(p$,4+p*7+4) 

9B29 1580 return 

676B 1590 rem ***************************************** 

2751 1600 rem *** wait for key and return key value *** 

9947 1610 rem ***************************************** 

24DE 1620 poke198,0 

E21C 1630 getk$:on-(k$="")goto1630:k=asc(k$) 

C489 1640 return 

6A28 1650 rem ********************************************* 

FCBF 1660 rem *** clear screen (bottom-up line removal) *** 

8AEE 1670 rem ********************************************* 

C863 1680 fori=jto1step-1:poke781,i:sys59903:next 

13D9 1690 return 

335D 1700 rem ******************************************************* 

E84B 1710 rem *** copy char rom and create some custom characters *** 

3AE8 1720 rem ******************************************************* 

8F4F 1730 poke56334,peek(56334)and254:poke1,peek(1)and251:poke781,2:poke782,255 

7425 1740 poke90,0:poke91,209:poke88,0:poke89,57:sys41960:poke1,peek(1)or4 

7898 1750 poke56334,peek(56334)or1:fori=14336to14343:reada:pokei,a:next 

C6C6 1760 fori=14552to14679:reada:pokei,a:next:fori=14800to14839:reada:pokei,a:next 

7259 1770 return 

E5EA 1780 data 60,126,189,126,24,24,60,0       :rem @ trophy cup 

0C1B 1790 data 254,254,254,238,254,254,254,0   :rem [ dice one 

CEA8 1800 data 254,250,254,254,254,190,254,0   :rem £ dice two 

14F5 1810 data 254,250,254,238,254,190,254,0   :rem ] dice three 

C8CE 1820 data 254,186,254,254,254,186,254,0   :rem ^ dice four (arrow up) 

3E53 1830 data 254,186,254,238,254,186,254,0   :rem < dice five (arrow left) 

10A6 1840 data 0,0,0,0,0,0,0,0                 :rem   space 

A122 1850 data 0,56,124,124,116,56,0,0         :rem ! dice dot 

7C23 1860 data 102,102,0,0,0,0,0,0             :rem " double quotes 

6AF4 1870 data 0,0,0,0,0,3,7,7                 :rem # dice top left corner 

78ED 1880 data 0,0,0,0,0,255,255,255           :rem $ dice top border 

8641 1890 data 0,0,0,0,0,128,192,192           :rem % dice top right corner 

2AB3 1900 data 7,7,7,7,7,7,7,7                 :rem & dice left border & 

F281 1910 data 192,192,192,192,192,192,192,192 :rem ' dice right border 

4367 1920 data 7,3,0,0,0,0,0,0                 :rem ( bottom left corner 

0E4B 1930 data 255,255,0,0,0,0,0,0             :rem ) dice bottom border 

4443 1940 data 192,128,0,0,0,0,0,0             :rem * bottom right corner 

8FCE 1950 data 254,186,254,186,254,186,254,0   :rem : dice six 

9D4C 1960 data 0,0,0,0,0,255,255,0             :rem ; table top border 

75AB 1970 data 0,0,0,0,0,255,255,192           :rem < table top limiter 

088F 1980 data 0,0,60,0,60,0,0,0               :rem = equal sign 

FC49 1990 data 56,68,68,254,254,254,254,254    :rem > lock symbol 

9B54 2000 rem ********************************************* 

7FF9 2010 rem *** initialize 4 sprites for yahtzee logo *** 

D9DA 2020 rem ********************************************* 

8A34 2030 pokev+23,0:pokev+78,0:pokev,156:pokev+1,76:pokev+2,180:pokev+3,76 

6FA5 2040 pokev+4,156:pokev+5,97:pokev+6,180:pokev+7,97 

1A86 2050 fori=.to3:pokev+39+i,13:next:pokev+27,255 

D4FB 2060 poke2040,232:poke2041,233:poke2042,234:poke2043,235 

A692 2070 fori=.to255:reada:poke14848+i,a:next 

95C7 2080 return 

7860 2090 rem *** sprite 0: yahtzee logo top left *** 

1824 2100 data 0,0,0,0,0,0,0,0,0,0,0,1,0,0,6,0 

2120 2110 data 0,9,0,0,23,0,0,47,0,0,95,0,0,95,0,0 

4B7C 2120 data 191,0,0,191,0,1,127,0,1,127,0,1,127,0,1,127 

CD0C 2130 data 0,1,127,0,1,126,0,1,125,0,1,253,0,0,12,13 

D5F0 2140 rem *** sprite 1 yahtzee logo top right *** 

9386 2150 data 0,15,0,0,127,192,63,255,240,195,255,216,62,127,244,252 

338B 2160 data 63,254,252,15,250,248,7,255,248,3,253,248,3,255,248,1 

88F3 2170 data 253,248,1,255,252,0,253,252,0,255,254,0,250,254,0,254 

20A5 2180 data 135,0,252,121,192,248,238,240,240,123,127,224,223,127,192,13 

494B 2190 rem *** sprite 2 yahtzee logo bottom left *** 

7189 2200 data 0,15,236,0,59,212,0,126,164,0,7,118,0,0,167,0 

0674 2210 data 0,247,0,0,161,0,0,240,15,0,160,31,128,128,54,192 

C7A8 2220 data 0,127,224,56,254,241,252,31,224,238,14,192,127,7,128,61 

CD6B 2230 data 3,0,47,0,0,63,0,0,61,0,0,47,0,0,56,13 

ACC3 2240 rem *** sprite 3 yahtzee logo bottom right *** 

73B3 2250 data 125,125,128,63,127,128,63,123,0,47,118,0,60,236,0,179 

CDA8 2260 data 216,0,207,224,0,127,128,0,0,0,0,0,0,0,15,224 

ED00 2270 data 0,3,248,0,3,232,0,2,248,0,3,184,0,3,248,0 

33A3 2280 data 3,232,0,2,248,0,3,128,0,0,0,0,0,0,0,13 

 


